

# The Riemann curvature tensor

```
{a,b,c,d,e,f,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\partial_{#}::PartialDerivative.

\Gamma^{a}_{b c}::Depends(\partial{#}).
\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2});

;::Symbol; # Suggsted by Kasper as a way to (possibly) make use of ; legal
# see https://cadabra.science/qa/473/is-this-legal-syntax?show=478
# this code works with and without this trick

# generic rule for first two covariant derivs of a downstairs-vector

deriv1 := A?_{a ; b} -> \partial_{b}{A?_{a}} - \Gamma^{c}_{a b} A?_{c}.
deriv2 := A?_{a ; b ; c} -> \partial_{c}{A?_{a ; b}}
      - \Gamma^{d}_{a c} A?_{d ; b}
      - \Gamma^{d}_{b c} A?_{a ; d}.

substitute (deriv2,deriv1)      # cdb (ex01, deriv2)

Mabc := M_{a ; b ; c}.          # cdb (ex02, Mabc)

substitute (Mabc,deriv2)      # cdb (ex03, Mabc)

distribute (Mabc)             # cdb (ex04, Mabc)
product_rule (Mabc)           # cdb (ex05, Mabc)

Macb := M_{a ; c ; b}.         # cdb (ex06, Macb)

substitute (Macb,deriv2)      # cdb (ex07, Macb)

distribute (Macb)             # cdb (ex08, Macb)
product_rule (Macb)           # cdb (ex09, Macb)

diff := @(Mabc) - @(Macb).     # cdb (ex10, diff)
```

```

sort_product      (diff)          # cdb (ex11, diff)
rename_dummies    (diff)          # cdb (ex12, diff)
canonicalise      (diff)          # cdb (ex13, diff)
sort_sum          (diff)          # cdb (ex14, diff)
factor_out        (diff,$M_{a?}$) # cdb (ex15, diff)

```

$$\begin{aligned}
A^?_{a;b;c} &\rightarrow \partial_c (\partial_b (A^?_a) - \Gamma^d_{ab} A^?_d) - \Gamma^d_{ac} (\partial_b (A^?_d) - \Gamma^e_{db} A^?_e) - \Gamma^d_{bc} (\partial_d (A^?_a) - \Gamma^e_{ad} A^?_e) \\
M_{a;bc} &= M_{a;b;c} = \partial_c (\partial_b M_a - \Gamma^d_{ab} M_d) - \Gamma^d_{ac} (\partial_b M_d - \Gamma^e_{db} M_e) - \Gamma^d_{bc} (\partial_d M_a - \Gamma^e_{ad} M_e) \\
M_{a;cb} &= M_{a;c;b} = \partial_b (\partial_c M_a - \Gamma^d_{ac} M_d) - \Gamma^d_{ab} (\partial_c M_d - \Gamma^e_{dc} M_e) - \Gamma^d_{cb} (\partial_d M_a - \Gamma^e_{ad} M_e) \\
M_{a;bc} - M_{a;cb} &= M_d (-\partial_c \Gamma^d_{ab} + \partial_b \Gamma^d_{ac} - \Gamma^e_{ab} \Gamma^d_{ce} + \Gamma^e_{ac} \Gamma^d_{be})
\end{aligned}$$

```

\begin{gather*}
\backslash\text{cdb}\{\text{ex01}\}\backslash\backslash
M_{\{a;bc\}} = \backslash\text{cdb}\{\text{ex02}\} = \backslash\text{cdb}\{\text{ex03}\}\backslash\backslash
M_{\{a;cb\}} = \backslash\text{cdb}\{\text{ex06}\} = \backslash\text{cdb}\{\text{ex07}\}\backslash\backslash[10pt]
M_{\{a;bc\}} - M_{\{a;cb\}} = \backslash\text{cdb}\{\text{ex15}\}
\end{gather*}

```